QT       += core gui declarative opengl

TARGET = box2ddemo
TEMPLATE = app

VERSION = 0.2

include(qml-box2d/box2d-static.pri)

SOURCES += main.cpp

OTHER_FILES += qml/*.qml


RESOURCES = box2ddemo.qrc

symbian {
    CONFIG += mobility
    MOBILITY += sensors

    HEADERS += accelerometerfilter.h

    TARGET = Box2DDemo
    ICON = icons/wheel.svg

    # To lock the application to landscape orientation
    LIBS += -lcone -leikcore -lavkon

    TARGET.EPOCHEAPSIZE = 0x100000 0x2000000
    TARGET.EPOCSTACKSIZE = 0x14000
}

unix:!symbian {
    maemo5 {
        CONFIG += mobility
        MOBILITY += sensors

        HEADERS += accelerometerfilter.h

        BINDIR = /opt/usr/bin
        DATADIR = /usr/share
        DEFINES += DATADIR=\\\"$$DATADIR\\\" \
                   PKGDATADIR=\\\"$$PKGDATADIR\\\"

        target.path = $$BINDIR

        desktop.path = $$DATADIR/applications/hildon
        desktop.files += $${TARGET}.desktop

        iconxpm.path = $$DATADIR/pixmap
        iconxpm.files += icons/xpm/$${TARGET}.xpm

        icon26.path = $$DATADIR/icons/hicolor/26x26/apps
        icon26.files += icons/26x26/$${TARGET}.png

        icon40.path = $$DATADIR/icons/hicolor/40x40/apps
        icon40.files += icons/40x40/$${TARGET}.png

        icon64.path = $$DATADIR/icons/hicolor/64x64/apps
        icon64.files += icons/64x64/$${TARGET}.png

        INSTALLS += desktop \
                    iconxpm \
                    icon26 \
                    icon40 \
                    icon64
    } else {
        target.path = /usr/local/bin
    }

    INSTALLS += target
}
